
DB2 - <offline> - Declaration view
"AlarmDB"     
Global data block DB 2
Name: Family:
Author: Version: 0.1

Block version: 2
Time stamp Code:

Interface:
02/21/2012 09:34:23 AM
02/21/2012 09:34:23 AM

Lengths (block/logic/data): 00128  00012  00000

Block: DB2  

Address Name Type Initial value Comment
 0.0 STRUCT
 +0.0   AlarmAll BOOL FALSE 报警汇总
 +0.1   AlarmResetFlag BOOL FALSE 报警复位标记
 +0.2   AlarmNewFlag BOOL FALSE 新报警标记
 +0.3   AlarmHornResetFlag BOOL FALSE 报警消声标记
 +2.0   AlarmNums INT 0 报警数量
 +4.0   AlarmBit ARRAY[1..32]
 *0.1   BOOL
 +8.0   AlarmFP ARRAY[1..32]
 *0.1   BOOL
=12.0 END_STRUCT

FC2 - <offline>
"AlarmFC"     
Name: Family:
Author: Version: 0.1

Block version: 2
Time stamp Code:

Interface:
02/21/2012 09:52:14 AM
02/21/2012 09:39:08 AM

Lengths (block/logic/data): 00362  00238  00010

 Name  Data Type  Address  Comment 

  IN 0.0

    AlarmDB Block_DB 0.0 报警DB

    AlarmCount Int 2.0 报警条目

    AlarmReset Bool 4.0 复位按钮

    HornReset Bool 4.1 消声按钮

    FlashBit Bool 4.2 报警闪烁位

  OUT 0.0

    AlarmLight Bool 6.0 报警指示灯

    AlarmHorn Bool 6.1 报警蜂鸣器

  IN_OUT 0.0

  TEMP 0.0

    LoopNum Int 0.0

    PT1 DWord 2.0

    PT2 DWord 6.0

  RETURN 0.0

    RET_VAL 10.0

Block: FC2  

Network: 1      

初始化

      SET   
      SAVE  

      OPN   #AlarmDB #AlarmDB           -- 报警DB

      L     P#4.0
      T     #PT1 #PT1              
      L     P#8.0

      T     #PT2 #PT2              

      L     0
      T     DBW    2
      SET   
      R     DBX    0.2

Network: 2      

      L     #AlarmCount #AlarmCount        -- 报警条目
NEXT: T     #LoopNum #LoopNum          

//计算总的报警条目
      A     DBX [#PT1] #PT1              
      S     DBX    0.0
      JNB   M001
      L     DBW    2
      L     1
      +I    
      T     DBW    2
M001: NOP   0

//如果没有报警，则复位报警汇总标记位
      L     DBW    2
      L     0
      ==I   
      R     DBX    0.0
      R     DBX    0.1
      R     DBX    0.2

//判断是否有新报警发生
      A     DBX [#PT1] #PT1              
      FP    DBX [#PT2] #PT2              
      JNB   M002
      SET   
      S     DBX    0.2
M002: NOP   0

      L     #PT1 #PT1              
      L     P#0.1
      +D    
      T     #PT1 #PT1              

      L     #PT2 #PT2              
      L     P#0.1
      +D    
      T     #PT2 #PT2              

      L     #LoopNum #LoopNum          
      LOOP  NEXT

Network: 3      

报警指示灯处理

//如果有报警发生并按下复位按钮则置位报警复位标记
      A     DBX    0.0
      A     #AlarmReset #AlarmReset        -- 复位按钮
      S     DBX    0.1

//报警指示灯输出
      O(    
      A     DBX    0.0
      AN    DBX    0.1
      A     #FlashBit #FlashBit          -- 报警闪烁位
      )     
      O(    
      A     DBX    0.0
      A     DBX    0.1
      )     
      =     #AlarmLight #AlarmLight        -- 报警指示灯

Network: 4      

报警蜂鸣器处理

//如果有报警发生并按下消声按钮则置位消声标记
      A     DBX    0.0
      A     #HornReset #HornReset         -- 消声按钮
      S     DBX    0.3
//如有新报警，则复位消声标记
      A     DBX    0.2
      R     DBX    0.3

//报警蜂鸣器输出
      A     DBX    0.0
      AN    DBX    0.3
      =     #AlarmHorn #AlarmHorn         -- 报警蜂鸣器


